
Calculating Program Memory Checksums Using a PIC16F87X
INTRODUCTION

Many applications require the microcontroller to cal-
culate a checksum on the program memory to deter-
mine if the contents have been corrupted.  Until now,
the only family of PICmicro®  microcontrollers to have
the capability to read from program memory are the
PIC17CXXX devices.  The PIC16F87X devices are
the first 14-bit core PICmicro microcontrollers  that
are able to access program memory in the same
fashion as used with data EEPROM memory.  These
devices are FLASH extensions of the popular
PIC16C7X family.  Table 1 shows a comparison
between the two PICmicro microcontroller families.

TABLE 1: PIC16C7X vs. PIC16F87X

ACCESSING MEMORY

The data EEPROM and FLASH Program memory are
both accessed using the same method.  An address
and/or data value are stored in Special Function Reg-
isters (SFR) and then memory is accessed using con-
trol bits in other SFRs.  There are six SFRs required to
access memory:

• EECON1

• EECON2

• EEDATA

• EEDATH

• EEADR

• EEADRH

When interfacing to data EEPROM memory, the
address is stored in the EEADR register and the data is
accessed using the EEDATA register.  The operation is
controlled using the EECON1 and EECON2 registers.  The
register map for EECON1 is shown in Figure 1.  EECON2
is not a physical register.  Reading it will result in all '0's.
This register is used exclusively in the EEPROM and
FLASH write sequences.

When interfacing to FLASH program memory, the
address is stored in the EEADRH:EEADR registers and
the data is accessed using the EEDATH:EEDATA regis-
ters.  Since the same set of control registers are used
to access data and program memory, the EEPGD bit
(EECON1<7>) is used to indicate to the microcontroller
whether the operation is going to be on data memory
(EEPGD = 0) or program memory (EEPGD = 1).  Refer to
Section 7.0 in the PIC16F87X data sheet (DS30292)
for more information about using the EEPROM and
FLASH memories.

Feature PIC16C7X PIC16C87X

Pins 28 or 40 28 or 40

Timers 3 3

Interrupts 11 or 12 13 or 14

Communica-
tion

PSP, USART, 

SSP (SPI or I2C 
Slave)

PSP, USART, 

SSP( SPI or I2C 
Master/Slave)

Frequency 20 MHz 20 MHz

A/D 8-bit 10-bit

CCP 2 2

Program Mem. 4K or 8K EPROM 4K or 8K FLASH

RAM 192 or 368 bytes 192 or 368 bytes

Data EEPROM None 128 or 256 bytes

Other --- In-Circuit Debugger

FIGURE 1: EECON1 REGISTER  

R/W-x U-0 U-0 U-0 R/W-x R/W-0 R/S-0 R/S-0

EEPGD — — — WRERR WREN WR RD R = Readable bit
W = Writable bit
S = Settable bit
U = Unimplemented bit, 

read as ‘0’
- n = Value at POR reset

bit7 bit0

bit 7: EEPGD: Program / Data EEPROM Select bit
1 = Accesses Program memory
0 = Accesses data memory
      Note: This bit cannot be changed while a write operation is in progress.

bit 6:4: Unimplemented: Read as '0'

bit 3: WRERR: EEPROM Error Flag bit
1 = A write operation is prematurely terminated
      (any MCLR reset or any WDT reset during normal operation)
0 = The write operation completed

bit 2: WREN: EEPROM Write Enable bit
1 = Allows write cycles
0 = Inhibits write to the EEPROM

bit 1: WR: Write Control bit
1 = initiates a write cycle. (The bit is cleared by hardware once write is complete. 
      The WR bit can only be set (not cleared) in software.
0 = Write cycle to the EEPROM is complete

bit 0: RD: Read Control bit
1 = Initiates an EEPROM read (read takes one cycle. RD is cleared in hardware. 
      The RD bit can only be set (not cleared) in software).
0 = Does not initiate an EEPROM read

HEX FILE FORMAT

Development tools from Microchip support the Intel
HEX Format (INHX8M), Intel Split HEX Format
(INHX8S), and the Intel HEX 32 Format (INHX32).  The
most commonly used formats are the INHX8M and the
INHX32.  These are the only formats discussed in this
document.  Please refer to Appendix A in the MPASM
User's Guide (DS33014) for more information about
HEX file formats.  The difference between INHX8M and
INHX32 is that INHX32 supports 32-bit addresses
using a linear address record.  The basic format of the
hex file is the same between INHX8M and INHX32 as
shown below:

:BBAAAATTHHHH...HHHHCC

Each data record begins with a 9 character prefix and
always ends with a 2 character checksum.  All records
begin with a ':' regardless of the format.  The individual
elements are described below.

• BB - is a two digit hexadecimal byte count repre-
senting the number of data bytes that will appear 
on the line.

• AAAA - is a four digit hexadecimal address repre-
senting the starting address of the data record. 
Format is high byte first followed by low byte, the 
address is doubled because this format only sup-
ports 8-bits (to find the real PICmicro address, 
simply divide the value AAAA by 2).

• TT - is a two digit record type that will be '00' for 
data records, '01' for end of file records and '04' 
for extended address record (INHX32 only).

• HHHH - is a four digit hexadecimal data word. For-
mat is low byte followed by high byte. There will be 
BB/2 data words following TT.

• CC - is a two digit hexadecimal checksum that is 
the two's complement of the sum of all the pre-
ceding bytes in the line record.

HEX File Preparation

The checksum used to verify program memory con-
tents is a 14-bit number calculated only on the program
memory contents of a HEX file.  The reason that only
14-bits is used is because the PIC16F87X has 14-bit
wide program memory.  

The first step to obtaining the checksum is to get a com-
plete HEX file that has all address locations specified.
This can be easily accomplished in MPLAB by enabling
the programmer, either PROMATE II or PICSTART
PLUS, whichever one is available.  Load the HEX file
into MPLAB using the menus File -> Import -> Down-
load to Memory.  Then save the HEX file using File ->
Export -> Save HEX File.  Make sure that the Program
Memory box is checked with a range of 0 to 8191 and
the Configuration bits and IDs box are also checked.  It
is optional to check the EEPROM memory box depend-
ing on you application.  This will create a complete HEX
file including all program memory, configuration word,
IDs, and optionally EEPROM memory.

The checksum provided by a programmer, such as
PROMATE II or PICSTART PLUS, is not valid because
the configuration word and device ID are included in the
calculation.  Therefore, a different program is required
to calculate the program memory checksum.  Once a
complete HEX file has been obtained by the previously
presented method, it must be processed and modified
to contain the checksum.  The program CHECKSUM.EXE,
which is a DOS based program, reads in the HEX file,
calculates the checksum, and outputs the new HEX file
with checksum included.  The checksum is calculated
by:

1. Adding together the memory locations 0x0000
to 0x1FFE.

2. Mask off all but the lower 14-bits.
3. Take the 2's complement of Step 2.
4. Mask off all but the lower 14-bits.
5. Save this value into the HEX file at address

0x1FFF.

The program ignores all configuration word, ID, and
EEPROM memory information in the HEX file and
dumps it to the output file unchanged.  The output file
can then be programmed into the PIC16F87X device.

PICmicro Code

The code used by the PIC16F87X to calculate check-
sum uses 36 words of program memory and two data
memory locations.  The example code uses data mem-
ory locations 0x7E and 0x7F to store the calculated
checksum.  These locations are shared across all
banks.  The user can optionally change these locations
and add banking into the routine.  Figure 2 shows the
flowchart for the routine.  The checksum is created
such that by adding up all program memory locations,
a 14-bit result of 0x0000 is obtained.  Since the calcu-
lation is done in 16-bits, the result will actually be
0x4000, but the upper two bits are masked off by the
routine.  Example 1 shows the code in MPASM to cal-
culate the program memory checksum. If the program
memory verifies, the routine returns a '1'. If a failure is
detected, the routine returns a '0'.

LISTING 1: PROGRAM MEMORY CHECKSUM ROUTINE
CalcChecksum

bsf STATUS,RP1 ;Go to Bank 2
bcf STATUS,RP0
clrf ChecksumL ;Clear the Checksum
clrf ChecksumH ;registers
clrf EEADR ;Set the Program Memory
clrf EEADRH ; address to 0x0000

CLoop ;Loop for each location
bsf STATUS,RP0 ; to read memory location
bsf EECON1,EEPGD ;Set for program memory
bsf EECON1,RD ;Set for read operation
bcf STATUS,RP0 ;Go to Bank 2
nop
movf EEDATA,W ;Add low byte to Checksum
addwf ChecksumL,F
btfsc STATUS,C ;Check for overflow
incf ChecksumH,F ;Yes, increment Checksum
movf EEDATH,W ;Add high byte
addwf ChecksumH,F
incf EEADR,F ;Increment low address
btfsc STATUS,Z ;Check for overflow
incf EEADRH,F ;Increment high address
movf EEADRH,F ;Check to see if
btfss STATUS,Z ; address wrapped
goto CLoop ; from 0x1fff to
movf EEADR,F ; 0x0000
btfss STATUS,Z
goto CLoop

;Checkcum calculation complete
bcf ChecksumH,7 ;Clear upper 2 bits
bcf ChecksumH,6 ; only 14-bit checksum
movf ChecksumH,F ;Checksum should be 0
btfss STATUS,Z
retlw0 ;Checksum failed
movf ChecksumL,F
btfss STATUS,Z
retlw0 ;Checksum failed
retlw1 ;Checksum passed
DS91026A-page  4   1998 Microchip Technology Inc.

FIGURE  1: FLOWCHART  

Calculate Checksum

Initialize Checksum
registers to 0x00

Initialize EEADRH:EEADR
registers to 0x0000

Z = 1?

Z = 1?

EEADRH:
EEADR = 0?

Mask off upper two
bits of Checksum

Checksum = 0?

Read a word out
of program memory

Add EEDATA to Checksum

Increment Checksum

Add EEDATH to Checksum

Increment EEADR

Increment EEADRH

Return 1 in W Return 0 in W

Yes

Yes

No

No

No

No

high byte

